

MEMOTECH GRAPHICS SCREENDUMP

This routine dumps a copy of the Memotech graphics screen to an Epson printer. It produces a double-sized copy of the original which can either be black on white or *vice versa*.

The easiest way to read the

graphics screen on the Memotech is to use Control-B. This, when sent to the screen, has the same effect as GR\$(x,y,n), and expects three more bytes where x,y are the coordinates of the point to be read and n is the number of bits to be read down.

Before the routine is called, VS4 or a graphics screen must be present or an error will occur.

Chris Amor

MTX TO EPSON SCREEN DUMP

The routine has four modes of operation.
The second byte holds the type in this case.
(0) This uses algorithm to match alternate points.
(1) Uses block of four points per screen pixel.
(2) Inverse of type 0.
(3) inverse of type 1.

written by Chris Amor. July 1984.

50 CODE

```
864B      LD A,0          ;TYPE OF PRINT
864D      EX AF,AF
864E      LD D,191       ;TOP OF DUMP
8650      LD B,0         ;BOT
8652      LD E,0         ;LEFT
8654      LD C,255       ;RIGHT
8656      LD IX,£FD75    ;PRORPL (OUTPUT DEVICE)
865A      LD HL,£03FC    ;ADJUST SIZE OF DUMP
865D      LD A,B
865E      AND L
865F      AND 191
8661      LD B,A
8662      LD A,D
8663      OR H
8664      AND 191
8666      LD D,A
```


```

8667      LD HL, BYTE
866A      LD (HL), 0 ; CLEAR FIRST POSITION
866C      PUSH BC
866D      POP HL
866E      LD (IX+0), 1 ; SELECT PRINTER
8672      RST 10
8673      DB £83, 27, "A", 8 ; SET LINE FEED FOR EPSON
8677      LD B, D
8678      JR INSTART
867A ROW:  DEC B ; SET UP NEXT ROW POINTER
867B      DEC B
867C      DEC B
867D      DEC B
867E INSTART: PUSH BC
867F      LD A, L
8680      SUB E
8681      LD B, 0
8683      LD C, A
8684      INC BC
8685      SLA C
8687      RL B
8689      RST 10 ; SELECT EPSON GRAPHICS MODE
868A      DB £A5, 10, 13, 27, "*", 4, £C0 ; FOR (BC) BYTES
8691      POP BC
8692      LD C, E
8693 COL:  LD (IX+0), 0 ; SELECT SCREEN
8697      RST 10
8698      DB £A2, 27, "C", £E0, £B1, 4 ; GR$(X, Y, 4) RESULT IN WKAREA
869E      LD A, (£FE1A) ; BYTE RETURNED BY GR$
86A1      PUSH BC
86A2      CALL ABYTE ; DOUBLE SIZE
86A5      LD (IX+0), 1 ; SELECT PRINTER
86A7      RST 10
86AA      DB £C0 ; SEND BC TO PRINTER
86AB      POP BC
86AC      LD A, C
86AD      CP L
86AE      JR Z, FINISH ; IF LAST COLUMN
86B0      INC C ; NEXT COLUMN
86B1      JR COL
86B3 FINISH: LD A, B
86B4      SUB H
86B5      SUB 3
86B7      JR NZ, ROW ; FOR NEXT ROW
86B9 TIDYUP: RST 10
86BA      DB £B6, 10, 13, 27, "A", 10, 13 ; RESET LINE FEED
86C1      LD (IX+0), 0 ; RESET SCREEN
86C5      RET ; TO BASIC ?
86C6 ABYTE: LD C, A ; ENTER WITH BYTE IN A
86C7      PUSH DE ; RESULT RETURNED IN BC
86C8      XOR A ; ROUTINE DOUBLES HEIGHT AND LENGTH OF FLOT
86C9      BIT 0, C
86CB      JR Z, NEXT1
86CD      OR 3
86CF NEXT1: BIT 1, C
86D1      JR Z, NEXT2
86D3      OR 12
86D5 NEXT2: BIT 2, C
86D7      JR Z, NEXT3
86D9      OR 48
86DB NEXT3: BIT 3, C
86DD      JR Z, NEXT4
86DF      OR 192
86E1 NEXT4: PUSH HL
86E2      EX AF, AF
86E3      BIT 1, A
86E5      LD H, 0
86E7      JR Z, NOINV ; JUMP FOR BLACK INK
86E9      LD H, £FF
86EB NOINV: BIT 0, A
86ED      JR NZ, TYPE1
86EF TYPE0: EX AF, AF ; MATCHES INTERMEDIATE POINTS
86F0      XOR H ; INVERSE BYTE IF H=£FF
86F1      LD B, A ; ALGORITHM FOR INTERMEDIATE BYTES
86F2      LD A, (BYTE)
86F5      LD E, A
86F6      AND B
86F7      LD D, A
86F8      LD A, E
86F9      SRL A
86FB      AND B
86FC      OR D
86FD      LD D, A
86FE      LD A, E
86FF      SLA A
8701      AND B
8702      OR D
8703      LD C, A
8704      LD A, B
8705      LD (BYTE), A
8708      POP HL
8709      POP DE
870A      RET ; BC HOLDS PAIR OF BYTES
870B TYPE1: EX AF, AF ; NO MATCHING
870C      XOR H ; INVERSE IF H=£FF
870D      LD B, A
870E      LD C, A
870F      POP HL
8710      POP DE
8711      RET ; B AND C HOLD SAME BYTE
8712 BYTE: DB 0
8713      RET
8714      RET

```

Symbols:

BYTE	8712	INSTART	867E
ROW	867A	COL	8693
ABYTE	86C6	FINISH	86B3
NEXT1	86CF	NEXT2	86D5
NEXT3	86DB	NEXT4	86E1
NOINV	86EB	TYPE1	870B
TYPE0	86EF	TIDYUP	86B9



Screen dump